# $Id$
# Maintainer: AndyRTR <andyrtr@archlinux.org>

pkgname=libwpd
pkgver=0.10.2
pkgrel=1
pkgdesc="Library for importing WordPerfect (tm) documents"
arch=('i686' 'x86_64')
url="https://libwpd.sourceforge.net/"
license=('LGPL')
depends=('gcc-libs' 'zlib' 'librevenge')
makedepends=('libgsf>=1.14.26' 'doxygen' 'boost')
options=('!emptydirs')
source=(https://downloads.sourceforge.net/sourceforge/${pkgname}/${pkgname}-${pkgver}.tar.xz)
sha1sums=('ad9d53e22e746b449f8a2937481e2b36766908b9')

build() {
  cd ${pkgname}-${pkgver}
  ./configure --prefix=/usr --disable-static
  make
}

check() {
  cd $pkgname-$pkgver
  make check
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR="${pkgdir}" install
}
